PRÁCTICA 2. CURSO 2016-2017. INTEGRALES MÚLTIPLES. CAMPOS ESCALARES Y VECTORIALES.

1 Integrales múltiples

Con wxMAXIMA es posible realizar integrales múltiples que se resuelvan aplicando el teorema de Fubini:no hay más que escribir el comando "integrate" (igual que para una variable). Veamos el siguiente ejemplo:

```
--> integrate(integrate(3*x*y,x,0,1),y,2,6);
(%o1) 24
```

o, cambiando el orden de integración

```
--> integrate(integrate(3*x*y,y,2,6),x,0,1);
(%o2) 24
```

En el ejemplo anterior el recinto de integración era rectangular, por lo que es sencillo expresar los intervalos donde varían x e y.
Cuando el recinto de integración no es tan sencillo nos toca a nosotros definir los límites concretos de la integral; para eso, pueden ayudarnos los dibujos que hace MAXIMA.

Veamos el siguiente ejemplo: queremos calcular la integral de la función f(x,y)=e^{x/y} en el recinto del primer cuadrante (x e y positivas) encerrado por las curvas x=y^3 y x=y^2. Buscamos en qué puntos se cortan las curvas:

```
--> solve(y^3=y^2,y);
(%o3) [y=0,y=1]
```

Pintamos la gráfica. Para ello previamente llamamos al paquete "draw" (también podemos hacerlo usando plot2d, como hemos visto en la práctica 0):

```
--> load(draw)$
    wxdraw2d(color=blue,implicit(y^3=x,x,0,1,y,0,1),color=red,
    implicit(y^2=x,x,0,1,y,0,1));
```

(%t5)



(%o5)

Observamos que hemos dibujado en azul la curva x=y^3, mientras que en rojo aparece la curva x=y^2. Esta gráfica nos ayuda a ver los límites de integración: Observamos que si "x" varía en [0,1], entonces "y" varía en [x^(1/3),x^(1/2)]; o más facilmente, si "y" varía en [0,1], entonces "x" varía en [y^3,y^2]. Si tomamos esta última opción, podemos integrar y obtenemos el resultado:

```
--> integrate(integrate(exp(x/y),x,y^3,y^2),y,0,1);
```

(%o6) $-\frac{\%e-3}{2}$

Veamos un ejemplo más. Queremos integrar la función x^2+y^2 en el recinto descrito de la siguiente manera
{(x,y)\ (x^2+y^2)^2 <= 4(x^2-y^2), x>=0}.
Si intentamos dibujar la gráfica en implícitas, vemos que nos va a salir complicado, e incluso wxMaxima tarda demasiado tiempo en realizar la representación:

```
--> load(draw)$
    wxdraw2d(color=blue,implicit((x^2+y^2)^2=4*(x^2-y^2),x,0,2,y,0,2);
```

Por ello, interrumpimos el programa, ya que observamos que un cambio a polares nos facilitaría el trabajo, pero wxMaxima no lo realiza automáticamente; es por ello por lo que debemos hacerlo nosotros.
El dominio, en polares, se puede describir como:
{r<= 2*sqrt(cos(2t))con t en el intervalo [-pi/4,pi/4]}

Si dibujamos la curva r= 2*sqrt(cos(2t)), se observa que t varia en el intervalo [-pi/4,pi/4]: (Hacemos la gráfica como hemos visto en el apartado3.2 de la Práctica 0)

```
(%i1) wxplot2d([2*(cos(2*ph)^(1/2))], [ph,-%pi/4,%pi/4],
      [plot_format, gnuplot],
      [gnuplot_preamble, "set polar;set size ratio 1; set zeroaxis;"],
      [x,0,3])$
```

(%t1)



Entonces la integral quedaría:
(la función a integrar es, x^2+y^2, que en polares es r^2; por tanto integraremos r^3, ya que tenemos que multiplicar por el jacobiano r):

```
(%i2) integrate(integrate(r^3,r,0,2*sqrt(cos(2*t))),t,-%pi/4,%pi/4);
```

defint: upper limit of integration must be real; found $2\sqrt{\cos(2\,t)}$
-- an error. To debug this try: debugmode(true);

Nota:puede daros un error o preguntaros si el cos(2*t) es positivo, negativo o cero. Contestad que positivo; el error se debe a que no puede manejar números imaginarios como límites de integración. Una solución sencilla es cambiar la anterior por la instrucción siguiente, donde el coseno va en valor absoluto:

```
(%i3) integrate(integrate(r^3,r,0,2*sqrt(abs(cos(2*t)))),
      t,-%pi/4,%pi/4);
```

(%o3) π

2 Campos Escalares y Vectoriales

Veamos ahora algunas instrucciones sencillas para trabajar con cálculo vectorial. Comenzamos cargando el paquete adecuado ('vect'):

```
(%i4) load(vect)$
```

Vamos ahora a definir un campo escalar, al que llamamos f,
y que es una función de 3 variables x, y, z:

```
(%i5) f(x,y,z):= y^2+z^2-x;
```

(%o5) $\mathrm{f}(x,y,z):=y^2+z^2-x$

Le pedimos a Maxima su gradiente y después su gradiente en (2,3,-1):

```
(%i6) gradf:[diff(f(x,y,z),x),diff(f(x,y,z),y),diff(f(x,y,z),z)];
```

(%o6) $[-1,2\,y,2\,z]$

```
(%i7) at(gradf,[x=2,y=3,z=-1]);
```

(%o7) $[-1,6,-2]$

Notemos que hemos calculado el gradiente definiéndolo como un vector cuyas componentes son las derivadas parciales de la función escalar f.

El gradiente puede calcularse de manera más rápida usando grad(f):

```
(%i8) express(grad(f(x,y,z)));
```

(%o8) $[\frac{d}{d\,x}(z^2+y^2-x),\frac{d}{d\,y}(z^2+y^2-x),\frac{d}{d\,z}(z^2+y^2-x)]$

aunque para que lo muestre como estamos habituados a verlo haremos:

```
(%i9) ev(%, diff);
```

(%o9) $[-1,2\,y,2\,z]$

Otra opción para calcular el gradiente es usar el paquete 'linearalgebra' que contiene el comando jacobian (así de paso aprendemos como se puede calcular la matriz jacobiana):

```
(%i10) load(linearalgebra)$
       jacobian([f(x,y,z)],[x,y,z]);
       at(%,[x=2,y=3,z=-1]);
```

(%o11) $\begin{bmatrix} -1 & 2\,y & 2\,z \end{bmatrix}$

(%o12) $\begin{bmatrix} -1 & 6 & -2 \end{bmatrix}$

También podemos calcular la matriz hessiana de f:

```
(%i13) hessian(f(x,y,z),[x,y,z]);
```

(%o13) $\begin{bmatrix} 0 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}$

o el Laplaciano (d^2/dx^2 + d^2/dy^2 + d^2/dz^2) de un campo escalar:

```
(%i14) express(laplacian(f(x,y,z)));
       ev(%,diff);
```

(%o14) $\frac{d^2}{d\,z^2}(z^2+y^2-x)+\frac{d^2}{d\,y^2}(z^2+y^2-x)+\frac{d^2}{d\,x^2}(z^2+y^2-x)$

(%o15) 4

Definamos ahora un campo vectorial

```
(%i16) F: [x*y,x^2*z,exp(x+y)];
```

(%o16) $[x\,y,x^2\,z,\%e^{y+x}]$

Vamos a calcular su divergencia:

```
(%i17) div(F);
```

(%o17) $\mathrm{div}([x\,y,x^2\,z,\%e^{y+x}])$

```
(%i18) express(%);
       ev(%, diff);
```

(%o18) $\frac{d}{d\,y}(x^2\,z)+\frac{d}{d\,z}\%e^{y+x}+\frac{d}{d\,x}(x\,y)$

(%o19) y

Exactamente igual, se calcula el rotacional, usando el comando "curl" en lugar de "div".

```
(%i20) express(curl(F));
       ev(%,diff);
```

(%o20) $[\frac{d}{d\,y}\%e^{y+x}-\frac{d}{d\,z}(x^2\,z),\frac{d}{d\,z}(x\,y)-\frac{d}{d\,x}\%e^{y+x},\frac{d}{d\,x}(x^2\,z)-\frac{d}{d\,y}(x\,y)]$

(%o21) $[\%e^{y+x}-x^2,-\%e^{y+x},2\,x\,z-x]$

3 Instrucciones para la entrega de las prácticas

- Las prácticas hay que hacerlas en fichero Maxima (extensión .wxm) y remitirlas por email.
- Aconsejable realizar un fichero para cada práctica (llamandolo, por ejemplo, Practica 1-nombre alumno.wxm)
- En los ficheros ir explicando lo que se va a realizar, incluyendo los comentarios que sean necesarios, como en el ejemplo de la sección anterior, y que siempre van en la forma:
/*COMENTARIO QUE SE QUIERE HACER*/
- Comprobar que no hay erratas y las sentencias escritas son correctas.

4 Un ejemplo resuelto

Lo que viene a continuación es una muestra de lo que hay que remitir en el fichero de prácticas. Copiar en primer lugar el enunciado y poner la resolución del ejercicio a continuación (observar como se van introduciendo los comentarios que aclaran lo que estamos realizando en la resolución).

PROBLEMA N°. Calcular las siguientes integrales dobles, dibujando previamente los recintos de integración:
a) f(x,y) = sqrt(4*x^2-y^2) en la región comprendida entre x=1, y=0, y=x.
b) f(x,y) = y/(x^2+y^2), en la región comprendida entre los círculos x^2+y^2>=2, x^2+y^2<=2*y. (Ejercicio 7 del examen de Sept 14, de Mat I).

```
-->  /*RESOLUCIÓN 1.a*/
     /*Realizamos la representación gráfica del recinto de
     integración*/

     wxplot2d([x, ['parametric, 1, t, [t, 0, 2], [nticks, 300]],
     ['parametric, t, 0, [t, 0, 2], [nticks, 300]]], [x,0,1.2],
     [y,-0.5,1.5])$
```

(%t22)




```
-->  /*Ahora ya podemos calcular la integral doble viendo donde
     varia cada variable: 0<x<1; 0<y<x*/

     integrate(integrate(sqrt(4*x^2-y^2),y,0,x),x,0,1);
Is  x  positive, negative, or zero?p;
```

(%o23) $\frac{2\pi+3^{3/2}}{18}$

```
(%i24) /*RESOLUCIÓN 1.b*/
       remvalue(all)$
       load(draw)$
       wxdraw2d(color=red,implicit(x^2+y^2=2,x,-2,2,y,0,2),
       color=blue,implicit(x^2+y^2=2*y,x,-2,2,y,1,2));
```

(%t26)



(%o26)

```
-->  /*El recinto de integración es la intersección de ambos
     círculos (exterior a la circunferencia roja e interior a
     la azul)*/

     /*Vamos a realizar un cambio a coordenadas polares, donde
     se observa que el ángulo t varía entre %pi/4 y 3*%pi/4,
     mientras que el radio r varía entre 2^0.5 y 2*/

     /*En primer lugar introducimos la función a integrar*/

     f(x,y):= y/(x^2+y^2);
```

(%o27) $f(x,y):=\frac{y}{x^2+y^2}$

```
-->  /* Hacemos el cambio a coordenadas polares e introducimos
     el jacobiano*/

     x:r*cos(t);
     y:r*sin(t);
     J:r;
```

(%o28) $r\cos(t)$

(%o29) $r\sin(t)$

(%o30) r

```
-->  /*La nueva función a integrar, en polares, será*/

     f(x,y)*J;
```

(%o31) $\frac{r^2\sin(t)}{r^2\sin(t)^2+r^2\cos(t)^2}$

```
(%i32) /*que si la simplificamos (utilizamos la simplicación
       trigonométrica que aparece en el menú superior de Maxima:
       Simplificar->Simplificación trigonométrica->
       Simplificar trigonometría)*/

       trigsimp(%);
```

(%o32) $\sin(t)$

```
-->  /*Por tanto, la integral a calcular será:*/

     integrate(integrate(%,r,2^0.5,2),t,%pi/4,3*%pi/4);
```

rat: replaced 0.5857864376269 by 22619537/38613965 = 0.58578643762691
rat: replaced 1.414213562373095 by 22619537/15994428 = 1.414213562373097
rat: replaced 0.5857864376269 by 22619537/38613965 = 0.58578643762691

(%o33) $0.5857864376269\sqrt{2}$

5 Ejercicios a entregar

PROBLEMA 1.
Representando previamente los recintos de integración,
se pide:

a) Calcular la integral doble de la función f(x,y)=(x^2+y^2)^(-3/2)
en el recinto D comprendido por x^2+y^2<=1, y>=x, x+y>=1.

b) La integral triple de
f(x,y,z)=(exp((x^2+y^2)^(1/2)))/(x^2+y^2)^(1/2)
en V, siendo V el volumen limitado inferiormente por el paraboloide
z=x^2+y^2, y superiormente por el plano z=4.

(ES EL EJERCICIO 5 DEL EXAMEN DE SEPT 2016 DE MATEMÁTICAS I)

PROBLEMA 2.
Considerar los campos vectoriales
F: [A*x*y,B*x^2*z,exp(C*x+D*y)] y G:(gradiente de x*y*z*tan(y)),
siendo A la última cifra no nula del NIF del alumno,
B la penúltima, ...,
y el campo escalar f:x^2*z*cos(y).
Se pide:

a) Calcular la divergencia de (F+G) y la suma de divergencia
de F y divergencia de G. ¿coinciden ambos valores?

b) Probar que rot(grad(f))es el vector nulo.

c) Probar que se verifica div(rot(F))=0.